

PROGRAMA PARA CÁLCULOS PETROQUÍMICOS (NORMA E VALORES DE NIGGLI) EM COMPUTADOR

Por

GILBERTO AMARAL (1)

SUMMARY

A Fortran program for computer calculation of the norm and Niggli values is presented. The calculation time, including input of data and output of results, is of about 35 seconds for each rock sample. The program was tested for rocks with calculations already made by hand, and the results agreed fairly well.

SUMARIO

É apresentado um programa Fortran para o cálculo da norma e dos valores de Niggli em computador. O tempo de cálculo, para cada amostra de rocha, é de cerca de 35 segundos, incluindo-se o tempo de leitura dos dados e perfuração dos cartões com os resultados. Testes efetuados com análises de rochas calculadas manualmente, forneceram resultados bastante concordantes.

INTRODUÇÃO

O presente trabalho foi desenvolvido com a finalidade de reduzir o tempo empregado nos cálculos petroquímicos pelos métodos convencionais. Além disso, os resultados são mais seguros, pois a possibilidade de erro fica restrita à perfuração dos cartões de dados e pode ser facilmente corrigida.

Cerca de 500 análises químicas de rochas foram coletadas na literatura ou fornecidas por colegas interessados, e calculadas segundo o programa ora apresentado. Para várias das amostras já contávamos com os cálculos efetuados manualmente, e os resultados obtidos no computador concordaram plenamente. Na Tabela I, apresentamos a comparação entre os resultados obtidos em três amostras da Ilha de Trindade, estudada por Almeida (1961).

O cálculo completo, para cada amostra, toma cerca de 35 segundos se incluirmos o tempo de leitura dos dados e perfuração dos

cartões com os resultados. Desta maneira, com o tempo que se leva para calcular manualmente uma amostra (cerca de 3 horas), calculam-se cerca de 180 amostras no computador, contando-se o tempo de perfuração dos cartões com os dados.

Para o cálculo da norma, o programa segue o método CIPW, com as modificações propostas por Niggli (Barth, 1952). Assim, obtemos a porcentagem molecular, e não em peso, dos minerais normativos. A sequência dos cálculos segue o método CIPW original, que é um tanto diferente daquela proposta por Barth (1952). Vitaliano, Harvey e Cleveland (1965) apresentaram um programa semelhante, somente que sem o cálculo dos valores de Niggli. Um fluxograma dos cálculos, fornecido por esses autores foi utilizado para a programação inicial. Todavia, ele se mostrou falho para o cálculo de rochas insaturadas, obrigando-nos a reformular grande parte do programa original.

UTILIZAÇÃO DO PROGRAMA



Um cartão de processamento possui 80 colunas e 12 linhas, sendo que cada símbolo (número, letra, sinal aritmético, etc.) ocupa uma coluna ao ser perfurado. A entrada dos dados é feita por intermédio de um único cartão para cada amostra. As quatro primeiras colunas são reservadas para o número de identificação da amostra (de 0000 a 9999), e cada quatro colunas seguintes são reservadas para as porcentagens em peso dos diversos óxidos, obtidas na análise química. Para a perfeita utilização do programa deve ser observada rigorosamente a seguinte disposição

(1) Departamento de Engenharia de Minas da Escola Politécnica da Universidade de São Paulo.

para os resultados das análises: SiO_2 ; TiO_2 ; Al_2O_3 ; Fe_2O_3 ; FeO ; MnO ; MgO ; CaO ; Na_2O ; K_2O ; P_2O_5 ; CO_2 ; S; ZrO_2 ; Cr_2O_3 . Vejamos o seguinte exemplo:

Amostra nº 0012 (Monchiquito da tabela I)

$\text{SiO}_2 = 37,40$; $\text{TiO}_2 = 2,50$; $\text{Al}_2\text{O}_3 = 18,58$; $\text{Fe}_2\text{O}_3 = 3,96$; $\text{FeO} = 8,26$; $\text{MnO} = 0,15$; $\text{MgO} = 7,93$; $\text{CaO} = 10,00$; $\text{Na}_2\text{O} = 3,60$; $\text{K}_2\text{O} = 1,90$; $\text{P}_2\text{O}_5 = 0,65$; $\text{CO}_2 = 0,00$; S = 0,00; $\text{ZrO}_2 = 0,00$; $\text{Cr}_2\text{O}_3 = 0,00$.

No cartão de dados seria perfurado o seguinte:

00123740025018580396082600150793100003600
190006500000000000000000

As quatro primeiras colunas referem-se ao número da amostra e cada grupo de quatro colunas restantes a uma porcentagem em peso de um óxido (ou enxofre), que deve conter sempre quatro algarismos, dois inteiros e

dois fracionários, não sendo necessária a perfuração do ponto decimal (vírgula). Os zeros não necessitam ser perfurados.

Internamente, o programa é dividido em sete fases:

1 — Entrada — Leitura e preparação dos dados para o cálculo;

2 — Cálculo dos valores de Niggli;

3 — Cálculo dos minerais normativos acessórios;

4 — Cálculo dos minerais normativos félsicos provisórios;

5 — Cálculo dos minerais normativos máficos provisórios;

6 — Cálculo dos minerais normativos félsicos e máficos deficientes em sílica, se a rocha fôr insaturada, a partir dos correspondentes provisórios;

7 — Saída — Perfuração dos cartões com os resultados.

Tabela I — Comparação entre cálculos feitos manualmente e em computador

Rocha	Olivina Nefelinito (7) (1)		Nefelinito (8)		Monchiquito (12)	
	Manual	Computador	Manual	Computador	Manual	Computador
Norma(2)						
OR	7,50	7,375	5,00	4,330	0,00	0,430
AB	0,50	0,741	6,50	6,925	0,00	0,000
AN	21,25	21,554	19,00	19,585	30,00	30,129
NE	18,90	18,770	18,90	18,750	20,10	20,197
LC					8,80	9,316
KF					0,30	0,000
DI	27,40	27,098	24,80	24,777	14,80	13,695
OL	9,75	9,697	7,10	9,137	16,85	17,263
HM				0,037		
MT	8,25	8,192	8,50	8,384	4,35	4,311
IL	3,60	3,653	5,00	4,959	3,60	3,626
AP	2,90	2,917	3,20	3,190	1,30	1,415
Valôres de Niggli						
Si	75,40	75,54	76,00	76,10	77,50	77,59
Ti	3,80	3,77	5,30	5,20	3,80	3,90
Al	17,00	17,06	17,00	17,13	22,60	22,71
Fm	48,20	48,09	48,60	48,65	45,20	45,29
C	26,70	26,69	25,30	25,29	22,60	22,22
Alc	8,10	8,14	9,10	8,91	9,70	9,75
K	0,19	0,18	0,12	0,10	0,25	0,25
Mg	0,52	0,52	0,53	0,52	0,55	0,54
Qz	— 57,00	— 57,05	— 60,40	— 59,56	— 61,30	— 61,43

(1) Os números entre parênteses referem-se ao nº da amostra no trabalho de Almeida (1961).
(2) Porcentagem de minerais normativos moleculares.

No presente programa, a saída é feita por meio de cinco cartões, com a seguinte disposição (por exemplo, para o monchiquito de Trindade):

AMOSTRA Nº 0012

0.000	0.043	0.000	30.129	20.197	9.316	0.000			
13.695	0.000	0.000	0.000	17.263	0.000		22.796	7.069	
0.000	0.000	0.000	1.415	0.000	3.626	0.000	0.000	4.311	0.000
77.59	3.90	22.71	45.29	22.22	9.75	0.25	0.54	— 61.43	

No primeiro cartão sai perfurado o número da amostra, no segundo a porcentagem molecular dos minerais normativos félsicos, na seguinte ordem: Qz, Or, Ab, An, Ne, Lc e Kf. No terceiro, os minerais normativos máficos, na seguinte disposição: Di, En, Hi, Wo, Ol, Ac, espaço, En*, Fs. Estes dois últimos valores serão explicados adiante. O quarto cartão contém os valores dos minerais acesórios, na seguinte ordem: Calcita, zircão, cromita, apatita, pirita, ilmenita, titanita, coríndon, magnetita e hematita. Finalmente, no quinto cartão estão perfurados os valores de Niggli, na seguinte sequência: Si, Ti, Al, Fm, C, Alk, K, Mg e Qz.

Por convenção, metasilicato de potássio, metasilicato de sódio, casilita e perovskita, aparecem nas posições correspondentes à albita, anortita, enstatita e hiperstênio, respectivamente, com sinal negativo. Por exemplo, onde aparecer $An = -3.50$, leia-se $Ns = 3.50$.

Os valores En* e Fs que aparecem isolados no terceiro cartão, servem para o cálculo das moléculas do Di e da Ol. No exemplo acima, a metade do valor do diopsídio refere-se à wolastonita e a outra metade à enstatita + ferrosilita. Devido que, numa certa

fase dos cálculos, o computador decide se calcula enstatita ou hiperstênio, é desejável saber-se a composição molecular deste último. Assim, $Hi = En + Fs$, e estes dois va-

lores são os que aparecem isolados. Pode acontecer que somente a enstatita seja calculada, neste caso ela aparece ao lado do diopsídio e os valores isolados serão nulos. Isto foi feito porque, após o cálculo dos minerais máficos deficientes em sílica, poderemos querer saber qual a composição molecular destes minerais ($Di = (En + Fs) + Wo$ e $Ol = Fa + Fo$). Voltando ao exemplo, $En + Fs = 6,847$, mas $En^*/Fs = 3,22$, donde $Fs = 1,61$, $En = 5,24$ e $Wo = 6,847$. Para o caso da olivina, $Fo + Fa = 17,263$, mas $Fo/Fa = En^*/Fs = 3,22$, portanto: $Fa = 4,09$ e $Fo = 13,17$.

Segue-se, abaixo, o programa propriamente dito. Ele consta de 257 cartões de processamento, perfurados, cada um correspondendo a uma instrução em linguagem Fortran. Ao copiar o programa em cartões de processamento, deve-se tomar o cuidado de não perfurar a coluna nº 6, de manter exatamente a mesma sequência das instruções e de reproduzir todos os símbolos existentes. Todavia, dadas as peculiaridades de cada computador (que aceite linguagem Fortran), recomendamos aos interessados que procurem orientação do pessoal especializado antes do início dos trabalhos.

P R O G R A M A

C CALCULOS PETROQUIMICOS (NORMA E VALORES DE NIGGLI)

C PROGRAMA FORTRAN POR GILBERTO AMARAL EPUSP

DIMENSION A(15), B(15), C(10), D(10), PC(15), PP(15)

B(1) = 60.09
B(2) = 79.90
B(3) = 50.98
B(4) = 79.85
B(5) = 71.85
B(6) = 70.94
B(7) = 40.32
B(8) = 56.08

B(9) = 30.99
B(10) = 47.10
B(11) = 70.98
B(12) = 44.01
B(13) = 32.06
B(14) = 123.22
B(15) = 76.01

```

1 READ 90,K,(A(I),I=1,15)

90 FORMAT (I4,15F4.2)
   DO 5 I=1,15

5 PP(I)=A(I)*1000./B(I)
  SOMA=0.
  DO 6 I=1,15

6 SOMA=SOMA+PP(I)
  DO 7 I=1,15

7 PC(I)=(PP(I)*100.)/SOMA
  C(1)=60.09
  C(2)=79.90
  C(3)=101.96
  C(4)=159.70
  C(5)=71.85
  C(6)=70.94
  C(7)=40.32
  C(8)=56.08
  C(9)=61.98
  C(10)=94.19
  DO 200 I=1,10

200 D(I)=(A(I)*1000.)/C(I)
  D(5)=D(5)+2.*D(4)
  SOMA=D(3)+D(5)+D(6)+D(7)+D(8)+D(9)+D(10)
  AL=(100.*D(3))/SOMA
  FM=(100.*(D(5)+D(6)+D(7)))/SOMA
  CA=(100.*D(8))/SOMA
  ALK=(100.*(D(9)+D(10)))/SOMA
  SI=(D(1)*100.)/SOMA
  TI=(D(2)*100.)/SOMA
  AK=D(10)/(D(10)+D(9))
  AMG=D(7)/(D(5)+D(6)+D(7))
  IF(AL-ALK)201,202,202

201 X=(100.+3.*AL+ALK)
  GO TO 203

202 X=100.+4.*ALK

203 QTZ=SI-X
  PC(5)=PC(5)+PC(6)
  CALC=2.*PC(12)
  PC(8)=PC(8)-PC(12)
  AP=8.*PC(11)/3.
  PC(8)=PC(8)-5.*PC(11)/3.
  ZR=2.*PC(14)
  PC(1)=PC(1)-PC(14)
  CM=3.*PC(15)/2.
  PC(5)=PC(5)-PC(15)/2.
  PI=3.*PC(13)/2.
  PC(5)=PC(5)-PC(13)/2.
  IF(PC(5)-PC(2))8,9,9

8 AILM=2.*PC(5)
  PC(2)=PC(2)-PC(5)
  TIT=3.*PC(2)
  PC(8)=PC(8)-PC(2)
  PC(5)=0.
  PC(1)=PC(1)-PC(2)
  GO TO 10

9 AILM=2.*PC(2)
  PC(5)=PC(5)-PC(2)
  TIT=0.

10 IF(PC(3)-PC(10))11,12,12

```

```

11 OR=5.*PC(3)
   PC(10)=PC(10)-PC(3)
   PC(1)=PC(1)-3.*PC(3)
   PC(3)=0.
   AB=-3.*PC(10)/2.
   PC(1)=PC(1)-PC(10)/2.
   PC(10)=0.
   GO TO 13

12 OR=5.*PC(10)
   PC(3)=PC(3)-PC(10)
   PC(1)=PC(1)-3.*PC(10)

13 IF(PC(3)-PC(9))14,18,18

14 AB=5.*PC(3)
   PC(9)=PC(9)-PC(3)
   PC(1)=PC(1)-3.*PC(3)
   PC(3)=0.
   IF(PC(4)-PC(9))15,16,16

15 AC=4.*PC(4)
   PC(9)=PC(9)-PC(4)
   PC(1)=PC(1)-2.*PC(4)
   PC(4)=0.
   AN=-3.*PC(9)/2.
   PC(1)=PC(1)-PC(9)/2.
   PC(9)=0.
   GO TO 17

16 AC=4.*PC(9)
   PC(4)=PC(4)-PC(9)
   PC(1)=PC(1)-2.*PC(9)
   PC(9)=0.

17 AN=0.
   GO TO 20

18 AB=5.*PC(9)
   PC(3)=PC(3)-PC(9)
   PC(1)=PC(1)-3.*PC(9)
   AC=0.
   IF(PC(3)-2.*PC(8))19,21,21

19 AN=5.*PC(3)/2.
   PC(8)=PC(8)-PC(3)/2.
   PC(1)=PC(1)-PC(3)
   PC(3)=0.

20 WO=2.*PC(8)
   PC(1)=PC(1)-PC(8)
   CR=0.
   GO TO 22

21 AN=5.*PC(8)
   PC(3)=PC(3)-2.*PC(8)
   CR=PC(3)
   WO=0.

22 IF(PC(4)-2.*PC(5))23,25,25

23 AMT=3.*PC(4)/2.
   PC(5)=PC(5)-PC(4)/2.
   HM=0.
   PC(4)=0.
   IF(PC(5))26,26,24

24 AEN=2.*PC(7)
   PC(1)=PC(1)-PC(7)
   FS=2.*PC(5)
   PC(1)=PC(1)-PC(5)
   HI=AEN+FS
   EN=0.
   GO TO 27

25 AMT=3.*PC(5)
   PC(4)=PC(4)-2.*PC(5)
   PC(5)=0.
   HM=PC(4)

26 EN=2.*PC(7)
   PC(1)=PC(1)-PC(7)
   AEN=0.
   FS=0.
   HI=0.

27 IF(PC(1))28,58,58

28 IF(EN+HI-WO)29,32,32

29 DI=2.*(HI+EN)
   HI=0.
   EN=0.
   OL=0.
   WO=WO-DI/2.
   IF(WO/4.+PC(1))30,31,31

30 EN=-3.*WO/4.
   PC(1)=PC(1)+WO/4.
   WO=0.
   GO TO 40

31 WO=WO+4.*PC(1)
   EN=3.*PC(1)
   PC(1)=0.
   QZ=0.
   GO TO 59

32 DI=2.*WO
   WO=0.
   QZ=0.
   IF(EN)33,33,36

33 HI=HI-DI/2.
   IF(HI/4.+PC(1))34,35,35

34 OL=3.*HI/4.
   PC(1)=PC(1)+HI/4.
   HI=0.
   GO TO 40

35 OL=-3.*PC(1)
   HI=HI+4.*PC(1)
   PC(1)=0.
   GO TO 40

36 EN=EN-DI/2.
   IF(EN/4.+PC(1))37,38,38

37 OL=3.*EN/4.
   PC(1)=PC(1)+EN/4.
   EN=0.
   GO TO 40

38 OL=-3.*PC(1)
   EN=EN+4.*PC(1)
   PC(1)=0.

40 IF(PC(1))41,100,100

100 QZ=0.
   GO TO 59

41 IF(PC(1)+TIT/3.)42,42,57

42 HI=-2.*TIT/3.
   PC(1)=PC(1)+TIT/3.
   TIT=0.
   IF(AB)47,47,43

```

```

43 IF(2.*AB/5.+PC(1))44,45,45
44 ANE=3.*AB/5.
   PC(1)=PC(1)+2.*AB/5.
   AB=0.
   GO TO 47
45 ANE=3.*PC(1)/2.
   AB=AB+5.*PC(1)/2.
   PC(1)=0.
47 IF(PC(1))48,110,110
110 QZ=0.
   GO TO 60
48 IF(OR/5.+PC(1))49,50,50
49 ALC=4.*OR/5.
   PC(1)=PC(1)+OR/5.
   OR=0.
   GO TO 51
50 ALC=4.*PC(1)
   OR=OR+5.*PC(1)
   PC(1)=0.
51 IF(PC(1))52,120,120
120 QZ=0.
   GO TO 61
52 IF(DI/4.+PC(1))53,56,56
53 EN=EN-3.*DI/8.
53 OL=OL+3.*DI/8.
   PC(1)=PC(1)+DI/4.
   DI=0.
   IF(ALC/4.+PC(1))54,55,55
54 AKF=3.*ALC/4.
   PC(1)=PC(1)+ALC/4.
   ALC=0.
   QZ=0.
   GO TO 62
55 AKF=3.*PC(1)
   ALC=ALC+4.*PC(1)
   PC(1)=0.
81 QZ=0.
   GO TO 62
56 EN=EN+3.*PC(1)/2.
   OL=OL-3.*PC(1)/2.
   DI=DI+4.*PC(1)
   PC(1)=0.
   QZ=0.
   GO TO 61
57 HI=2.*PC(1)
   TIT=TIT+3.*PC(1)
   QZ=0.
   GO TO 59
58 QZ=PC(1)
   DI=0.
   OL=0.
59 ANE=0.
60 ALC=0.
61 AKF=0.
62 PUNCH 63,K
63 FORMAT (11HAMOSTRA NO., 15)
   PUNCH 64,QZ,OR,AB,AN,ANE,ALC,AKF
64 FORMAT (7F8.3)
   PUNCH 65,DI,EN,HI,WO,OL,AC,AEN,FS
65 FORMAT (6F8.3,16X,2F8.3)
   PUNCH 66,CALC,ZR,CM,AP,PI,AJLM,TIT,CR,AMT,HM
66 FORMAT (10F8.3)
   PUNCH 70,SI,TI,AL,FM,CA,ALK,AK,AMG,QTZ
70 FORMAT (9F8.2)
   GO TO 1
   END

```

AGRADECIMENTOS

Desejamos expressar aqui nossa gratidão ao pessoal do Centro de Cálculo Eletrônico do Instituto de Pesquisas Matemáticas da Universidade de São Paulo, em especial aos Engenheiros Tomasz Kowaltowski e Imre Simon, pela orientação recebida.

BIBLIOGRAFIA

- ALMEIDA, F. F. M. de (1961) — Geologia e Petrologia da Ilha de Trindade. Monografia XVIII da Divisão de Geologia e Mineralogia D. N. P. M. 197 pp.
- BARTH, T. E. W. (1952) — Theoretical Petrology. John Wiley & Sons, N. York, 387 pp.
- VITALIANO, C. J.; HARVEY, R. D. e CLEVELAND, J. H. (1965) — Computer Program for Norm Calculation. Am. Miner. vol. 50, pp. 495-498.